
HC-SR04   -   Module de détection aux ultrasons   -   Utilisation avec Picaxe 

 
Le capteur HC-SR04 utilise les ultrasons pour déterminer la distance d'un objet. Il offre une excellente plage 
de détection sans contact, avec des mesures de haute précision et stables. Son fonctionnement n'est pas 
influencé par la lumière du soleil ou des matériaux sombres, bien que des matériaux comme les vêtements 
puissent être difficiles à détecter. 
 

Caractéristiques 

 Dimensions : 45 mm x 20 mm x 15 mm 

 Plage de mesure :  2 cm à 400 cm 

 Résolution de la mesure :  0.3 cm 

 Angle de mesure efficace :  15 ° 

 Largeur d'impulsion sur l'entrée de déclenchement :  10 µs  
(Trigger Input Pulse width) 

 

Broches de connection 

 Vcc =  Alimentation +5 V DC 

 Trig =  Entrée de déclenchement de la mesure (Trigger input) 

 Echo =  Sortie de mesure donnée en écho (Echo output) 

 GND =  Masse de l'alimentation 
 

Spécifications et limites 

Paramètre Min Type Max Unité 

Tension d'alimentation 4.5 5.0 5.5 V 

Courant de repos 1.5 2.0 2.5 mA 

Courant de fonctionnement 10 15 20 mA 

Fréquence des ultrasons - 40 - kHz 

Attention : la borne GND doit être connectée en premier, avant l'alimentation sur Vcc. 

 
Branchement 

Le branchement du capteur HC-SR04 sur un 
microcontrôleur PICAXE est des plus simple 
(le schéma ci-contre illustre l'utilisation d'un 
circuit 08M2). 

Si les deux éléments sont alimentés en 5 V 
(ou 4.5 V), il suffit de relier les bornes 
d'alimentations +/- entre elles. 

L'entrée "Trig" du capteur doit être reliée à 
une sortie du PICAXE. 

Quant à la sortie "Echo" du capteur, il faut 
la relier à une entrée du PICAXE qui peut 
aussi être programmée en sortie si on désire 
mettre en œuvre l'exemple 2 de 
programme ci-dessous. 

 

 

 

http://www.google.ch/imgres?imgurl=http://microcontrollerelectronics.com/wp-content/uploads/2014/10/HCSR04.jpg&imgrefurl=http://microcontrollerelectronics.com/distance-sensing/&h=473&w=800&tbnid=0NU54tqe8wbBrM:&docid=uojwh4kBN3X1eM&ei=xYxYVpG7EofuUKnAiMgB&tbm=isch&ved=0ahUKEwjRsfbPirHJAhUHNxQKHSkgAhkQMwgnKAIwAg

Fonctionnement 

Pour déclencher 
une mesure, il faut 
présenter une im-
pulsion "high" (5 V) 
d'au moins 10 µs 
sur l'entrée "Trig". 
Le capteur émet 
alors une série de 8 
impulsions ultraso-
niques à 40 kHz, 
puis il attend le 
signal réfléchi. 
Lorsque celui-ci est 
détecté, il envoie 
un signal "high" sur la sortie "Echo", dont la durée  est proportionnelle à la distance mesurée. 
 
Distance de l'objet 

La distance parcourue par un son se calcule en multipliant la vitesse du son, environ 340 m/s (ou 
34'000 cm/1'000'000 µs) par le temps de propagation, soit :   d = v · t     (distance = vitesse · temps) 
 
Le HC-SR04 donne une durée d'impulsion en dizaines de µs. Il faut donc multiplier la valeur obtenue par 10 µs 
pour obtenir le temps t. On sait aussi que le son fait un aller-retour. La distance vaut donc la moitié. 
 
d = 34'000 cm/1'000'000 µs · 10us · valeur / 2      en simplifiant      d =170'000 /1'000'000  cm · valeur 
 

Finalement,      d = 17/100 cm · valeur   

 
La formule  d = durée/58 cm figure aussi dans le manuel d'utilisation du HC-SR04 car la fraction 17/1000 est 
égale à 1/58.8235. Elle donne cependant des résultats moins précis. 
 
Note : A grande distance, la surface de l'objet à détecter doit mesurer au moins 0.5 m² 
 
Exemple 1 de programme 

 

' Essai du capteur HC-SR04 à ultrasons  -  28.11.2015 L.Bachelard 

' =============================================================== 

 

  symbol  trig  = C.1             ' sortie du Picaxe sur Trig du capteur 

  symbol  echo  = C.2             ' entrée sur Picaxe de Echo du capteur (note: pulsin 

                                  ' requiert une entrée sous la forme C.3 et non pinC.3) 

  symbol  calc  = w1              ' variable 16 bits, calcul du temps et de la distance 

 

#terminal 4800                    ' ouverture du terminal pour affichage des valeurs 

 

low trig                          ' programmation de Trig en mode sortie et mise à 0 

do                                ' boucle principale 

  pulsout trig,1                  ' impulsion sur Trig en dizaines de us (1 => 10us) 

  pulsin  echo,1,calc             ' lit la durée de l'impulsion Echo (1 = flanc montant) 

  let     calc = calc*17/100      ' calcul de la distance 

  sertxd  ("d = ",#calc," cm",cr,lf)  ' affichage de la distance sur le terminal 

  pause   1000                    ' attente de 1s 

loop                              ' boucle sans fin 

 

 
Exemple 2 de programme 

Cette variante de programme est destinée à corriger le fait que, quand le capteur ne rencontre aucun 
obstacle, ou lorsque son émetteur ou son récepteur ultrasonique est occulté (par un doigt par exemple), ou 
encore lorsque l'objet à détecter absorbe le rayon ultrasonique (c'est le cas des vêtements en laine), le HC-
SR04 se bloque parfois et délivre sur sa sortie "Echo" un signal qui reste à 0. 

Dans ce cas, il suffit de lui appliquer un signal "bas" pendant 100 ms sur sa sortie "Echo". Cette fonction paraît 
étonnante, mais elle a apparemment l'effet de décharger un condensateur sur le circuit du capteur et de 
réinitialiser son fonctionnement. Il semblerait aussi que ce défaut ne se trouve que sur les HC-SR04 bon 
marché. 
 
' Essai 2 du capteur HC-SR04 à ultrasons  -  28.11.2015 L.Bachelard 

' ================================================================= 

 

' Avec correction du HC-SR04 si la distance lue reste bloquée à 0 

 

' J'ai trouvé sur internet le truc pour faire repartir le capteur : 

' il faut lui appliquer une impulsion "low" de 100ms sur sa sortie "Echo" 

 

  symbol  trig  = C.1             ' sortie du Picaxe sur Trig du capteur 

  symbol  echo  = C.2             ' entrée sur Picaxe de Echo du capteur (note: pulsin 

                                  ' requiert une entrée sous la forme C.3 et non pinC.3) 

  symbol  calc  = w1              ' variable 16 bits, calcul du temps et de la distance 

 

#terminal 4800                    ' ouverture du terminal pour affichage des valeurs 

 

low trig                          ' programmation de Trig en mode sortie et mise à 0 

do                                ' boucle principale 

  pulsout trig,1                  ' impulsion sur Trig en dizaines de us (1 => 10us) 

  pulsin  echo,1,calc             ' lit la durée de l'impulsion Echo (1 = flanc montant) 

  if calc = 0 then                ' si 0 (capteur bloqué), alors 

    sertxd ("Erreur, distance 0 !",cr,lf)  ' affichage erreur 

    reverse echo                  ' inversion de l'entrée Echo en mode sortie 

    low   echo                    ' mise à 0 de la borne Echo 

    pause 100                     ' pendant 100ms 

    reverse echo                  ' puis, passage à nouveau en mode entrée 

  endif                           ' fin de la condition 

  let     calc = calc*17/100      ' calcul de la distance 

  sertxd  ("d = ",#calc," cm",cr,lf)  ' affichage de la distance sur le terminal 

  pause   1000                    ' attente de 1s 

loop                              ' boucle sans fin 

 

 
Conclusion 

Le capteur HC-SR04 est intéressant. Pour un coût très bas, il donne des résultats étonnants de précision. 
L'écart est d'environ 3 cm avec un objet placé à 2 m, ce qui représente une erreur inférieure à 2 %. 
 

Sources 

 User's Manual du HC-SR04 

 Recherches sur Internet 

 Essais personnels 

 
 
Le 28 novembre 2015   -   Lucien Bachelard   -   lu.bachelard@bluewin.ch 


